Symbolic Programming by Example

Thomas Hahn

Max-Planck-Institut für Physik
München

https://feynarts.de/lectures/sym.pdf

https://feynarts.de/lectures/sym.tar.gz

List of Examples

- **Antisymmetric Tensor**
  Built-in in FORM, easy in Mathematica.
- **Application of Momentum Conservation**
  Easy in Mathematica, complicated in FORM.
- **Abbreviationing**
  Easy in Mathematica, new in FORM.
- **Simplification of Color Structures**
  Different approaches.
- **Calculation of a Fermion Trace**
  Built-in in FORM, complicated in Mathematica.
- **Tensor Reduction**

Reference Books, Formula Collections

- V.I. Borodulin et al.
  CORE (Compendium of Relations)
  hep-ph/9507456, arXiv:1702.08246.



- Herbert Pietschmann
  Formulae and Results in Weak Interactions
  Springer (Austria) 2nd ed., 1983.



- Andrei Grozin
  Using REDUCE in High-Energy Physics
  Cambridge University Press, 1997.



Antisymmetric Tensor

The **Antisymmetric Tensor in n dimensions** is denoted by $\varepsilon_{i_1 i_2 \ldots i_n}$. You can think of it as a matrix-like object which has either -1, 0, or 1 at each position.

For example, the **Determinant** of a matrix, being a **completely antisymmetric** object, can be written with the ε-tensor:

$$\det A = \sum_{i_1,\ldots,i_n=1}^{n} \varepsilon_{i_1 i_2 \ldots i_n} A_{i_1 1} A_{i_2 2} \cdots A_{i_n n}$$

In practice, the ε-tensor is usually contracted, e.g. with vectors. We will adopt the following notation to avoid dummy indices:

$$\varepsilon_{\mu\nu\rho\sigma} p^\mu q^\nu r^\rho s^\sigma = \varepsilon(p, q, r, s)\,.$$

Antisymmetric Tensor in Mathematica

```
Eps[___, p_, ___, p_, ___] := 0

   (* implement linearity: *)

Eps[a___, p_Plus, b___] := Eps[a, #, b]&/@ p

Eps[a___, n_?NumberQ r_, b___] := n Eps[a, r, b]

   (* otherwise sort the arguments into canonical order: *)

Eps[args__] := Signature[{args}] Eps@@ Sort[{args}] /;
  !OrderedQ[{args}]
```



Momentum Conservation

Problem: Proliferation of terms in expressions such as

$$d = \frac{1}{(p_1 + p_2 - p_3)^2 + m^2}$$
$$= \frac{1}{p_1^2 + p_2^2 + p_3^2 + 2p_1p_2 - 2p_2p_3 - 2p_1p_3 + m^2},$$

whereas if $p_1 + p_2 = p_3 + p_4$ we could have instead

$$d = \frac{1}{p_4^2 + m^2}.$$

In Mathematica: just do `d /. p1 + p2 - p3 -> p4`.
Problem: FORM cannot replace sums.

Momentum Conservation in FORM

Idea: for each expression x, add and subtract a zero, i.e. form

$$\{x, y = x + 0, z = x - 0\}, \quad \textbf{where e.g.} \quad 0 = p_1 + p_2 - p_3 - p_4\,,$$

then select the shortest expression. But: how to select the shortest expression (in FORM)?

Solution: add the number of terms of each argument, i.e.

$$\{x, y, z\} \to \{\overset{1}{x}, \overset{2}{y}, \overset{3}{z}, \overset{4}{n_x}, \overset{5}{n_y}, \overset{6}{n_z}\}\,.$$

Then sort n_x, n_y, n_z, but when exchanging n_a and n_b, exchange also a and b:

```
symm 'foo' (4,1) (5,2) (6,3);
```

This unconventional sort statement is rather typical for FORM.

Momentum Conservation in FORM

```
#procedure Shortest(foo)

id `foo'([x]?) = `foo'([x], [x] + `MomSum', [x] - `MomSum');

* add number-of-terms arguments
id `foo'([x]?, [y]?, [z]?) = `foo'([x], [y], [z],
  nterms_([x]), nterms_([y]), nterms_([z]) );

* order according to the nterms
symm `foo' (4,1) (5,2) (6,3);

* choose shortest argument
id `foo'([x]?, ?a) = `foo'([x]);

#endprocedure
```



Abbreviationing

One of the most powerful tricks to both reduce the size of an expression and reveal its structure is to substitute subexpressions by new variables.

The essential function here is `Unique` with which new symbols are introduced. For example,

```
Unique["test"]
```

generates e.g. the symbol `test1`, which is guaranteed not to be in use so far.

The `Module` function which implements lexical scoping in fact uses `Unique` to rename the symbols internally because Mathematica can really do dynamical scoping only.

Abbreviationing in Mathematica

```
$AbbrPrefix = "c"

abbr[expr_] := abbr[expr] = Unique[$AbbrPrefix]

   (* abbreviate function *)
Structure[expr_, x_] := Collect[expr, x, abbr]

   (* get list of abbreviations *)
AbbrList[] := Cases[DownValues[abbr],
  _[_[_[f_]], s_Symbol] -> s -> f]

   (* restore full expression *)
Restore[expr_] := expr /. AbbrList[]
```



Abbreviationing in FORM

```
* collect w.r.t. some function

b Den;
.sort
collect acc;

* introduce abbreviations for prefactors

toPolynomial onlyfunctions acc;
.sort

* print abbreviations & abbreviated expr

#write "%X"
print +s;
```



Color Structures

In Feynman diagrams four types of Color structures appear:

Natural Representation



$\sim T^a_{ij} =$ SUNT$[a, i, j]$



$\sim T^a_{ij} T^a_{k\ell} =$ SUNTSum$[i, j, k, \ell]$

Adjoint Representation



$\sim f^{abc} =$ SUNF$[a, b, c]$



$\sim f^{abx} f^{xcd} =$ SUNF$[a, b, c, d]$

Unified Notation

The SUNF's can be converted to SUNT's via

$$f^{abc} = 2\mathrm{i}\left[\mathrm{Tr}(T^c T^b T^a) - \mathrm{Tr}(T^a T^b T^c)\right].$$

We can now represent all color objects by just SUNT:

- $\mathtt{SUNT}[i,j] = \delta_{ij}$
- $\mathtt{SUNT}[a,b,\ldots,i,j] = (T^a T^b \cdots)_{ij}$
- $\mathtt{SUNT}[a,b,\ldots,0,0] = \mathrm{Tr}(T^a T^b \cdots)$

This notation again avoids unnecessary dummy indices. (Mainly namespace problem.)

For purposes such as the "large-N_c limit" people like to use SU(N) rather than an explicit SU(3).

Fierz Identities

The Fierz Identities relate expressions with different orderings of external particles. The Fierz identities essentially express completeness of the underlying matrix space.

They were originally found by Markus Fierz in the context of Dirac spinors, but can be generalized to any finite-dimensional matrix space [hep-ph/0412245].

For SU(N) (color) reordering, we need

$$T^a_{ij} T^a_{k\ell} = \frac{1}{2}\left(\delta_{i\ell}\delta_{kj} - \frac{1}{N}\delta_{ij}\delta_{k\ell}\right).$$



Cvitanovich Algorithm

For an Amplitude:



- convert all color structures to (generalized) `SUNT` objects,
- simplify: apply Fierz identity on all internal gluon lines,
- expect `SUNT` with indices of external particles to remain.

For a Squared Amplitude:



- use the Fierz identity to get rid of all `SUNT` objects,
- expect `SUNT` to vanish, color factors (numbers) only.

For "hand" calculations, a pictorial version of this algorithm exists in the literature.

Color Simplify in FORM

```
* introduce dummy indices for the traces
repeat;
  once SUNT(?a, 0, 0) = SUNT(?a, DUMMY, DUMMY);
  sum DUMMY;
endrepeat;

* take apart SUNTs with more than one T
repeat;
  once SUNT(?a, [a]?, [b]?, [i]?, [j]?) =
    SUNT(?a, [a], [i], DUMMY) * SUNT([b], DUMMY, [j]);
  sum DUMMY;
endrepeat;

* apply the Fierz identity
id SUNT([a]?, [i]?, [j]?) * SUNT([a]?, [k]?, [l]?) =
  1/2 * SUNT([i], [l]) * SUNT([j], [k]) -
  1/2/(`SUNN') * SUNT([i], [j]) * SUNT([k], [l]);
```



Translation to Color-Chain Notation

In color-chain notation we can distinguish two cases:

a) Contraction of different chains:

$$\langle A| T^a |B\rangle \langle C| T^a |D\rangle = \frac{1}{2}\left(\langle A|D\rangle \langle C|B\rangle - \frac{1}{N}\langle A|B\rangle \langle C|D\rangle\right),$$

b) Contraction on the same chain:

$$\langle A| T^a |B| T^a |C\rangle = \frac{1}{2}\left(\langle A|C\rangle \operatorname{Tr} B - \frac{1}{N}\langle A| B |C\rangle\right).$$



Color Simplify in Mathematica

```
   (* same-chain version *)
sunT[t1___, a_Symbol, t2___, a_, t3___, i_, j_] :=
  (sunT[t1, t3, i, j] sunTrace[t2] -
    sunT[t1, t2, t3, i, j]/SUNN)/2

   (* different-chain version *)
sunT[t1___, a_Symbol, t2___, i_, j_] *
sunT[t3___, a_, t4___, k_, l_] ^:=
  (sunT[t1, t4, i, l] sunT[t3, t2, k, j] -
    sunT[t1, t2, i, j] sunT[t3, t4, k, l]/SUNN)/2

   (* introduce dummy indices for the traces *)
sunTrace[a__] := sunT[a, #, #]&[ Unique["col"] ]
```



Fermion Trace

Leaving apart problems due to γ_5 in d dimensions, we have as the main algorithm for the 4d case:

$$\begin{aligned}\operatorname{Tr}\gamma_\mu\gamma_\nu\gamma_\rho\gamma_\sigma\cdots = &+ g_{\mu\nu}\operatorname{Tr}\gamma_\rho\gamma_\sigma\cdots\\ &- g_{\mu\rho}\operatorname{Tr}\gamma_\nu\gamma_\sigma\cdots\\ &+ g_{\mu\sigma}\operatorname{Tr}\gamma_\nu\gamma_\rho\cdots\end{aligned}$$

This algorithm is recursive in nature, and we are ultimately left with

$$\operatorname{Tr}\mathbb{1} = 4\,.$$

(Note that this 4 is not the space-time dimension, but the dimension of spinor space.)

Fermion Trace in Mathematica

```
Trace4[mu_, g__] :=
Block[ {Trace4, s = -1},
  Plus@@ MapIndexed[
    ((s = -s) Pair[mu, #1] Drop[Trace4[g], #2])&,
    {g} ]
]

Trace4[] = 4
```



Tensor Reduction

The loop integrals corresponding to closed loops in a Feynman integral in general have a tensor structure due to integration momenta in the numerator. For example,

$$B_{\mu\nu}(p) = \int \mathrm{d}^d q \, \frac{q_\mu q_\nu}{\left(q^2 - m_1^2\right)\left((q-p)^2 - m_2^2\right)}.$$

Such tensorial integrals are rather unwieldy in practice, therefore they are reduced to linear combinations of Lorentz-covariant tensors, e.g.

$$B_{\mu\nu}(p) = B_{00}(p)\, g_{\mu\nu} + B_{11}(p)\, p_\mu p_\nu \,.$$

It is the coefficient functions B_{00} and B_{11} which are implemented in a library like LoopTools.

Tensor Reduction Algorithm

The first step is to convert the integration momenta in the numerator to an actual tensor, e.g. $q_\mu q_\nu \to N_{\mu\nu}$. FORM has the special command `totensor` for this:

```
totensor q1, NUM;
```

The next step is to take out $g_{\mu\nu}$'s in all possible ways. We do this in form of a sum:

$$N_{\mu_1 \ldots \mu_n} = \sum_{i=0,2,4,\ldots}^{n} \pi(0)^i \sum_{\substack{\text{all } \{\nu_1,\ldots,\nu_i\} \\ \in\{\mu_1,\ldots,\mu_n\}}} g_{\nu_1\nu_2} \cdots g_{\nu_{i-1}\nu_i} N_{\mu_1 \ldots \mu_n \setminus \nu_1 \ldots \nu_i}$$

The $\pi(0)^i$ keeps track of the indices of the tensor coefficients, i.e. it later provides the two zeros for every $g_{\mu\nu}$ in the index, as in D_{0012}.

Tensor Reduction Algorithm

To fill in the remaining $\pi(i)$'s, we start off by tagging the arguments of the loop function, which are just the momenta. For example:

$$C(p_1, p_2, \ldots) \to \tau\big(\pi(1)p_1 + \pi(2)p_2\big)\, C(p_1, p_2, \ldots)$$

The temporary function τ keeps its argument, the 'tagged' momentum p, separate from the rest of the amplitude.

Now add the indices of $N_{\mu_1 \ldots \mu_n}$ to the momentum in τ:

$$\tau(p)\, N_{\mu_i \ldots \mu_n} = p_{\mu_i} \cdots p_{\mu_n}\,.$$

Finally, collect all π's into the tensor-coefficient index.

Tensor Reduction in FORM

```
totensor q1, NUM;

* take out 0, 2, 4... indices for g_{mu nu}
id NUM(?b) = sum_(DUMMY, 0, nargs_(?b), 2,
  pave(0)^DUMMY * distrib_(1, DUMMY, dd_, NUM, ?b));

* construct tagged momentum in TMP
id C0i([p1]?, [p2]?, ?a) = TMP(pave(1)*[p1] + pave(2)*[p2]) *
  C0i(MOM([p1]), MOM([p2] - [p1]), MOM([p2]), ?a);

* expand momentum
repeat id TMP([p1]?) * NUM([mu]?, ?a) =
  d_([p1], [mu]) * NUM(?a) * TMP([p1]);

* collect the indices
chainin pave;
```



Tensor Reduction in Mathematica

```
tens[i_, p_][mu_, nu___] :=
Block[ {tens},
      (* take out g *)
  { MapIndexed[g[mu, #1] Drop[tens[{i,0,0}, p][nu], #2]&, {nu}],
      (* take out p *)
    (#1[mu] tens[{i,#2}, p][nu])&@@@ p }
]

tens[i_, _][] := C@@ Sort[Flatten[i]]


FindTensors[mu_, p_] :=
Block[ {tenslist},
  tenslist = tens[{}, MapIndexed[List, p]]@@ mu;
  Collect[Plus@@ Flatten[tenslist], _C]
]
```



More Complex Calculations

Often special requirements:

- Resummations (e.g. hbb in MSSM),
- Approximations (e.g. gaugeless limit),
- K-factors,
- Nontrivial renormalization.

Software design so far:

- Mostly 'monolithic' (one package does everything).
- Often controlled by parameter cards, not easy to use beyond intended purpose.
- May want to/must use other packages.

Example: $\mathcal{O}(\alpha_t^2)$ MSSM Higgs-mass corrections

Hollik, Paßehr 2014

Shopping List for the Diagrammatic Calculation:

① **Unrenormalized 2L self-energies**
$\Sigma^{(2)}_{hh}$, $\Sigma^{(2)}_{hH}$, $\Sigma^{(2)}_{hA}$, $\Sigma^{(2)}_{HH}$, $\Sigma^{(2)}_{HA}$, $\Sigma^{(2)}_{AA}$, $\Sigma^{(2)}_{H^+H^-}$
in gaugeless approximation at $p^2 = 0$ at $\mathcal{O}(\alpha_t^2)$.

② **1L diagrams with insertions of 1L counterterms.**

③ **2L counterterms** for ①.

④ **2L tadpoles** $T^{(2)}_h$, $T^{(2)}_H$, $T^{(2)}_A$ **at** $\mathcal{O}(\alpha_t^2)$ **appearing in** ③.

Template for Calculations

- Break calculation into several steps.
- Implement each step as independent program (invoked from command line).
- In lieu of 'in vivo' debugging keep detailed logs.
- Coordinate everything through a makefile.



Steps of the Calculation

Calculation split into 7 (8) steps:



Script Structure

- **Shell scripts (`/bin/sh`), run from command line as e.g.**
  `./1-amps arg1 arg2`
- `arg1` = `h0h0`, `h0HH`, `h0A0`, `HHHH`, `HHA0`, `A0A0`, `HmHp` **(self-energies),**
  `h0`, `HH`, `A0` **(tadpoles).**
- `arg2` = `0` **for virtual 2L diagrams,**
  `1` **for 1L diagrams with 1L counterterms.**
- **Inputs/outputs defined in first few lines, e.g.**

```
in=m/$1/2-prep.$2
out=m/$1/3-calc.$2
```

- **Symbolic output + log files go to '`m`' subdirectory.**
  Log file = Output file + `.log.gz`
- **Fortran code goes to '`f`' subdirectory.**

Step 0: Gaugeless Limit

Gaugeless approximation:

1. **Set gauge couplings** $g, g' = 0 \Rightarrow M_W, M_Z = 0$.
2. **Keep finite weak mixing angle.**
3. **Keep** $\frac{\delta M_W^2}{M_W^2}$ **and** $\frac{\delta M_Z^2}{M_Z^2}$ **finite.**

Must set $m_b = 0$ **so that** $\mathcal{O}(\alpha_t^2)$ **corrections form supersymmetric and gauge-invariant subset.**

Most efficient to modify Feynman rules (not ③, though):

- **Load** `MSSMCT.mod` **model file.**
- **Modify couplings, remove zero ones.**
- **Write out** `MSSMCTgl.mod` **model file.**

Step 1: Diagram Generation

- Generate 2L virtual and 1L+counterterm diagrams using wrappers for FeynArts functions.

Simple diagram selection functions, e.g.

```
sel[0][S[_] -> S[_]] = {
  t[3] && htb[6],
  t[3] && tb[6],
  t[3] && tb[6],
  t[3] && t[4] && htb[5],
  t[3] && htb[5|6],
  t[3] && htb[5],
  t[3] && t[5],
  t[5] && ht[3|4],
  t[3|4|5] && ht[3|4|5] }
```

one of $h_i, \tilde{\chi}, t, \bar{t}, b, \bar{b}$



Step 2: Preparation for Tensor Reduction

- Take $p^2 \to 0$ limit.
- Simplify ubiquitous sfermion mixing matrices U_{ij}, mostly by exploiting unitarity ($\sim$ 50% size reduction).



Efficiently Exploit Unitarity in Mathematica

Unitarity of 2 x 2 matrix: $UU^\dagger = U^\dagger U = \mathbb{1}$, **i.e.**

$$
\begin{aligned}
U_{11}U_{11}^* + U_{12}U_{12}^* &= 1, & U_{11}U_{21}^* + U_{12}U_{22}^* &= 0,\\
U_{21}U_{21}^* + U_{22}U_{22}^* &= 1, & U_{21}U_{11}^* + U_{22}U_{12}^* &= 0,\\
U_{11}U_{11}^* + U_{21}U_{21}^* &= 1, & U_{11}U_{12}^* + U_{21}U_{22}^* &= 0,\\
U_{12}U_{12}^* + U_{22}U_{22}^* &= 1, & U_{12}U_{11}^* + U_{22}U_{21}^* &= 0.
\end{aligned}
$$

Problem: `Simplify` **will rarely arrange the** U**'s in just the way that these rules can be applied directly.**

Solution: Introduce auxiliary symbols which immediately deliver the r.h.s. once `Simplify` **considers the l.h.s., i.e. increase the 'incentive' for** `Simplify` **to use the r.h.s.**

But: Upvalues work only one level deep.

Efficiently Exploit Unitarity in Mathematica

Introduce

$$\texttt{USf}[1, j]\ \texttt{USfC}[1, j] \to \texttt{UCSf}[1, j],$$

$$\texttt{USf}[2, j]\ \texttt{USfC}[2, j] \to \texttt{UCSf}[2, j],$$

$$\texttt{USf}[1, j]\ \texttt{USfC}[2, j] \to \texttt{UCSf}[3, j], \quad \textbf{+ ditto for 1}^{\textbf{st}}\ \textbf{index}$$

and formulate unitarity for the `UCSf`**:**

```
UCSf[2,1] = UCSf[1,2];    UCSf[3,2] = -UCSf[3,1];
UCSf[2,2] = UCSf[1,1];    UCSfC[3,2] = -UCSfC[3,1];
                          UCSf[2,3] = -UCSf[1,3];
...                       UCSfC[2,3] = -UCSfC[1,3];
```

Step 3: Tensor Reduction

- Relatively straightforward application of TwoCalc and FormCalc for tensor reduction.
- Observe: Need two Mathematica sessions since TwoCalc and FormCalc cannot be loaded into one session, easily accomodated in shell script.



Step 4: Simplification

- Tensor reduction traditionally increases # of terms most.
- Step 4 reduces size before combination of results.
- Empirical simplification recipe.
- 'DiagMark' trick (D. Stöckinger):
  - Introduce `DiagMark[`m_i`]` where m_i = masses in loop in FeynArts output.
  - Few simplifications can be made between parts with different `DiagMark` $\Rightarrow$ Can apply simplification as

    ```
    Collect[amp, _DiagMark, simpfunc]
    ```

  - Much faster.

Step 5: Calculation of Renormalization Constants

- Compute 1L renormalization constants (RC) with FormCalc.
- Substitute explicit mass dependence in

  $$\texttt{dMVsq1} \to \texttt{MV2 dMVsq1MV2} \quad (V = W, Z)$$

  such that gaugeless limit can be taken safely.
- Expand in ε, collect powers for easier handling later, e.g.

```
{ dMf1[3,3] -> RC[-1, dMf1[-1,3,3]] +
               RC[0, dMf1[0,3,3]],
```

expansion

```
{dMf1[-1,3,3] -> ...,
 dMf1[0,3,3] -> ...}
}
```

actual expressions for ε-coeffs

Step 6: Combination of Results

- Expand amplitude in ε (similar as RC).
- Insert RCs.
- Add genuine 2L counterterms (hand-coded).
- Pick only ε^0 term (unless debug flag set).
- Perform final simplification.



Step 7: Code Generation

- Introduce abbreviations to shorten code.
- Write out Fortran code using FormCalc's code-generation functions.
- Add static code which computes e.g. the necessary parameters for the generated code.
- Total final code size: 350 kBytes.

More details in arXiv:1508.00562.



Exercise

Devise programs in Mathematica and FORM which apply the Dirac equation to a suitably defined spinor, i.e.

$$\not{p}_i \left|u_i\right\rangle = m_i \left|u_i\right\rangle$$
$$\not{p}_i \left|v_i\right\rangle = -m_i \left|v_i\right\rangle$$

`https://feynarts.de/lectures/sym.pdf`
`https://feynarts.de/lectures/sym.tar.gz`